# $Id$
# Maintainer: AndyRTR <andyrtr@archlinux.org>
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=libxxf86vm
pkgver=1.1.4
pkgrel=1
pkgdesc="X11 XFree86 video mode extension library"
arch=('x86_64' 'i686')
license=('custom')
url="https://xorg.freedesktop.org/"
depends=('libxext' 'xf86vidmodeproto')
makedepends=('xorg-util-macros')
source=(${url}/releases/individual/lib/libXxf86vm-${pkgver}.tar.bz2{,.sig})
sha256sums=('afee27f93c5f31c0ad582852c0fb36d50e4de7cd585fcf655e278a633d85cd57'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # Alan Coopersmith <alan.coopersmith@oracle.com>

build() {
  cd libXxf86vm-${pkgver}
  ./configure --prefix=/usr --disable-static
  make
}

package()
{
  cd libXxf86vm-${pkgver}
  make DESTDIR="${pkgdir}" install

  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
